0>:48*-   #v_$>0145*p0>:045*p:48*1--#v_$7>44*>?<    _$++++\1-:#^_v >>^        >
 ^+1p*73\0:<          ^+1  _v#`2g*73:<>:2/ :v >0\2/:^           v< ^"TWANG.."<
145*g1+:145*p8 8*`     #v_ $>  >  44*>?<    _$++++:37*g2`!#v_    >
>7*p:37*g37*+p^vg*540+1g$#*73g<^$<   <>0\2/:^v _v#!`g*540: <    # >"rra ehT."^
^3\+1g*73::p*73<  >+2*+p# 045*^      ^       < <>:045*g64*1-g-#v_^   >:#,_$$  ^
C."*520<p*5400<   ^56g*73:g*540:_^#-g*46g*540:_^# -g*2+56g*540:<#     v"re A,B,
A("25*""v>5*p ^v1$<     >025*".niaga gniyrT .sevac gnikam pu dewercS"^>"iF :)CB
 A,B,C."<^  41 < _^#:p*7303p*37\5p*37\5p*37\5p*37\4p*37\4p*37\4$<v2"(abc): Move
*73g+2* 73:\g*73g+1*73:,*52.+1:g*540$_,#! #:<"Cave "0<$_,#! #:<*5<v30g*73g*46\g
540_v#- g-1*46<v"bats."*25_v#g*5440p*54:p*54:p*54:p*5450p*5440p*54<     v   4g*
  vv< > :045*g^>" raeh uoY">545*g!#v_25*".tfard "v>v #"Tunnels lead to:">#<6*g-
20_v$_^#!-g+1*73g*540:g*541$_,#! #:<"You feel a" <,:>$":A",,045*g:65+2*g1v^,:*5
  1>>025*"!supmuw a l" v>        `#v_v v_\$37*1++v^_^#v-1*46:,,"B:",*52.+<
2v^ $_,#! #:<"You smel"<2v` 2:-"A"$<   <!vg\g*540<    >g1+.":C"25*,,,64*g1+.">"
v>5*,,> ~:25*-!#v_:"a"-:^>#v_:01 vv0:<>`^>:045*p37*g: 4- #v_$025*"!tab tnaig a"
<>_#"#^^#      $<"0$       _v#!`-<> 1-^v$_,#! #:<"Yo"  "u've been snatched by "
"ni llef uoY...eeeeeeEEEEEE" "Y">:#,_$@0    >:2/: v> ^   <>5-#v_025*"!tip a ot"
\37*g5-#v_  $>  $$>  $$>$           >  >44*>?<    _$++++: :  045*g\:145*g\:37*g
" net"v">4-!#^_-!#^_-!#^_\0+# 45 *p#^ #1 #<v>0\2/:^#v541g *5\0< >"!supmuw a yb"
:#,_$@>"ae neeb ev'uoY...PLUG"  ^   >:#,_$^>    >  ^>*g-#^_025*>?>v>"llik uoY">
                            >37*1++v^"You bumped the wumpus."   ^#<^"ed the w"<
                                v40<^"The crafty wumpus dodged your arrow"  < "
                                5>0\25*\#v_    #^                     #<"!">?<u
                                *^-g*5<  >".gnihtyna tih t'ndid wo"vv_$^v"s"< "
                                >g\g14^>_v#"  >:2/ :v        "0<  v<,:  >"upm"^